

22. Zaawansowane formuły

Komputeryzacja sprawiła, że ogromną ilość danych zapisujemy cyfrowo. Rejestruje się w ten sposób w zasadzie każdą informację, a archiwa cyfrowe rozrastają się w błyskawicznym tempie. W związku z tym wciąż rośnie zapotrzebowanie na analizę danych. Wiele operacji na danych można wykonać za pomocą arkusza kalkulacyjnego. Warto nauczyć się zaawansowanych formuł, gdyż mogą się przydać w wielu dziedzinach aktywności zawodowej.

Cele lekcji

- Nauczysz się konstruować złożone formuły.
- Poznasz wybrane funkcje operujące na datach.
- Nauczysz się stosować funkcje tekstowe.
- Dowiesz się, jak korzystać z odwołań strukturalnych w tabelach Excela.
- Użyjesz funkcji WYSZUKAJ.PIONOWO.
- Nauczysz się używać formuł tablicowych.
- Zastosujesz metodę wstępującą przy tworzeniu złożonych formuł.

Funkcje arkusza kalkulacyjnego stosowane pojedynczo pozwalają rozwiązywać proste problemy. Funkcje można jednak łączyć, a to daje już naprawdę duże możliwości. W tym temacie pokażemy, jak tworzyć złożone formuły i do czego je wykorzystać. Omówimy też wybrane funkcje, które ułatwiają operowanie różnego typu danymi. Pokażemy ponadto, jak ułatwić sobie pracę w arkuszu kalkulacyjnym dzięki odwołaniom strukturalnym i formułom tablicowym.

22.1. Budowanie formuł z użyciem kilku funkcji

Kiedy rozwiązujemy jakiś problem, często musimy uwzględnić to, że otrzymamy różne wyniki w zależności od pewnych warunków. W programie Excel takie sterowanie wykonaniem formuły umożliwia funkcja JEŻELI. Wyobraź sobie, że chcesz utworzyć formułę, która sprawdza, czy liczba w komórce B1 jest różna od 0. W zależności od otrzymanego wyniku w komórce E1 ma się pojawić odpowiedni napis. Do utworzenia formuły w komórce E1 wykorzystamy właśnie funkcję JEŻELI.

Funkcja JEŻELI

Warto wiedzieć

W Excelu dostępne są funkcje logiczne ORAZ i LUB, które umożliwiają konstruowanie złożonych warunków.



Rys. 22.1. Formuła z użyciem funkcji JEŻELI

Pierwszy argument określa warunek, który sprawdzamy. W naszym wypadku – czy liczba zapisana w komórce B1 jest różna od 0. Drugi argument (oddzielony średnikiem) określa wynik funkcji, gdy sprawdzany warunek jest spełniony (wartość testu logicznego to PRAWDA). Trzeci argument określa wynik funkcji, gdy warunek nie jest spełniony (wartość testu logicznego to FAŁSZ). Rysunek 22.2 pokazuje, jak działa ta funkcja.

	A	B	C	D	E	F
1	Liczba:	0		Wynik:	zero	
2						

=JEŻELI(B1<>0;"różna od zera";"zero")

Rys. 22.2. Działanie funkcji JEŻELI wprowadzonej w komórce E1

Co jednak zrobić, jeśli chcielibyśmy dodatkowo sprawdzać, czy liczba jest dodatnia czy ujemna? Jeśli wiadomo, że liczba jest różna od zera, to wystarczy sprawdzić, czy jest większa od zera. W tym celu rozbudujemy naszą formułę. Ponownie zastosujemy funkcję JEŻELI, tym razem jako argument wcześniej użytej funkcji JEŻELI. Połączymy więc dwie funkcje, wykorzystując tzw. **zagnieżdżenie**. Przedstawia to rysunek 22.3.

• Zagnieżdżenie funkcji

	A	B	C	D	E	F
1	Liczba:	-7		Wynik:	ujemna	
2						

=JEŻELI(B1<>0;JEŻELI(B1>0;"dodatnia";"ujemna");"zero")

Rys. 22.3. Przykład zagnieżdżenia funkcji

Zwróć uwagę, że warunek $B1 > 0$ jest sprawdzany dopiero wtedy, gdy spełniony jest pierwszy warunek ($B1 \neq 0$). Jeżeli z kolei warunek $B1 > 0$ nie jest spełniony, to liczba w komórce B1 jest liczbą ujemną.

💡 Zapamiętaj

Funkcja zagnieżdżona to taka, która jest argumentem innej funkcji.

Zmodyfikujemy formułę z rysunku 22.3, aby sprawdzała również, czy dana liczba różna od zera jest parzysta. Chcemy, aby np. po wpisaniu w komórce B1 liczby 8, w E1 pojawił się napis „dodatnia parzysta”. Wykorzystamy w tym celu **funkcję MOD**, która zwraca resztę z dzielenia liczby przez dzielnik.

• Funkcja MOD

Liczba, której podzielność badamy

=MOD(B1;2)

Dzielnik

💡 Warto wiedzieć

Funkcja MOD zwraca wynik zgodny ze znakiem dzielnika, np. =MOD(-7;2) zwróci 1.

Rys. 22.4. Formuła z użyciem funkcji MOD

Warto wiedzieć

Parzystość liczby można również sprawdzić za pomocą funkcji **CZY.PARZYSTE** lub **CZY.NIEPARZYSTE**.

Funkcja określająca parzystość różnej od zera liczby z komórki B1 może wyglądać następująco: **JEŻELI(MOD(B1;2)=0;" parzysta"; " nieparzysta")**. W naszym przypadku badanie parzystości i znaku liczby są niezależne, gdyż znak nie ma wpływu na parzystość i na odwrót. Możemy więc połączyć wyniki sprawdzania za pomocą operatora łączenia napisów &:

JEŻELI(B1>0;"dodatnia";"ujemna")&JEŻELI(MOD(B1;2)=0;" parzysta";" nieparzysta")

Zauważ, że napisy, które ma zwracać funkcja sprawdzająca parzystość, celowo zapisaliśmy ze spacją na początku – dzięki temu słowa określające znak i parzystość będą rozdzielone, np. „dodatnia parzysta”. Kompletną formułę przedstawia rysunek 22.5.

	A	B	C	D	E	F
1	Liczba:	8		Wynik:	dodatnia parzysta	
2						

=JEŻELI(B1<>0;JEŻELI(B1>0;"dodatnia";"ujemna")&JEŻELI(MOD(B1;2)=0;" parzysta";" nieparzysta");"zero")

Rys. 22.5. Przykład zagnieżdżenia funkcji i łączenia tekstów

Dobra rada

Kiedy używasz bardzo złożonych formuł, lepiej zrezygnować z zagnieżdżania i stosować odwołania do innych komórek, które przechowują wyniki pośrednie. Zwiększa to czytelność i poprawia wydajność obliczeń.

Ćwiczenie 1

W nowym arkuszu w komórce B1 zapisz formułę, która będzie sprawdzać, czy liczba zapisana w komórce A1 jest podzielna przez 15. Jeśli jest, niech się wyświetli napis „Tak”, w przeciwnym razie – „Nie”. Skorzystaj z funkcji **JEŻELI**, **ORAZ** i **MOD**.

Obliczymy teraz, ile komórek we wskazanym zakresie spełnia wybrane kryterium, np. ilu jest pracowników w wieku 30–40 lat, na podstawie danych z zakresu komórek A1:B11 (rys. 22.6). Zaprezentujemy dwa sposoby rozwiązania tego zadania.

=LICZ.JEŻELI(B2:B11;"<=40")-LICZ.JEŻELI(B2:B11;"<30")

	A	B	C	D	E	F	G
1	ID pracownika	Wiek		Liczba pracowników w wieku 30-40 lat:			
2	P1	27					
3	P2	30				Sposób I	6
4	P3	40					
5	P4	49				Sposób II	6
6	P5	27					
7	P6	40					
8	P7	46					
9	P8	37					
10	P9	33					
11	P10	31					


=LICZ.WARUNKI(B2:B11;"<=40";B2:B11;">=30")

Rys. 22.6. Zliczanie komórek spełniających wybrane kryteria

W pierwszym rozwiązaniu w komórce G3 odejmujemy od siebie wyniki dwóch funkcji LICZ.JEŻELI. Pierwszym argumentem obu funkcji jest zakres komórek – w naszym przykładzie B2:B11. Dla każdej komórki z tego zakresu program sprawdza, czy liczba jest mniejsza lub równa 40, i zlicza wszystkie liczby spełniające ten warunek. Następnie odejmuje od nich liczby mniejsze niż 30 (zlicza je druga funkcja LICZ.JEŻELI).

W drugim rozwiązaniu wykorzystujemy funkcję LICZ.WARUNKI, która w odróżnieniu od funkcji LICZ.JEŻELI pozwala łączyć warunki. Dzięki temu nie musimy odejmować wyników, bo program od razu zlicza liczby spełniające jednocześnie oba warunki (rys. 22.7).

Zakres, dla którego sprawdzany jest pierwszy warunek



Rys. 22.7. Formuła z użyciem funkcji LICZ.WARUNKI

Ćwiczenie 2

W komórkach B2:B8 zapisz ceny jednostkowe wybranych urządzeń domowych, a w komórkach C2:C8 – ich liczbę. W komórce C10 oblicz sumę cen tych sprzętów, których jest mniej niż 3, a w komórce C11 sumę cen urządzeń, których jest 3 lub więcej. Skorzystaj z funkcji SUMA.JEŻELI.

22.2. Funkcje dotyczące dat

Obliczenia, w których wykorzystuje się daty, często są problematyczne. Trzeba w nich bowiem uwzględniać to, że co cztery lata jest rok przestępny, liczba dni w miesiącach jest różna, ale w każdym tygodniu stała.

W trakcie opracowywania danych z datami (np. tworzenia sprawozdań i rozliczeń) przydaje się automatyczne oznaczanie kwartału i roku, którego ten kwartał dotyczy. Wykonamy to zadanie w Excelu. W komórce B1 podajemy datę, w komórce H1 chcemy otrzymać rok, a w komórce E1 oznaczenie kwartału w postaci I, II, III lub IV.

	A	B	C	D	E	F	G	H	I
1	Data:	18.11.2018		Kwartał:	IV		Rok:	2018	
2									
3	Miesiąc:	11							
4									

=MIESIĄC(B1)
 =JEŻELI(B3<4;"I";JEŻELI(B3<7;"II";JEŻELI(B3<10;"III";"IV")))

=ROK(B1)

Rys. 22.8. Określanie roku i kwartału

Funkcja LICZ.JEŻELI, s. 330 [↗](#)

Funkcja LICZ.WARUNKI

Warto wiedzieć

W Excelu dostępna jest funkcja LICZ.PUSTE, która zlicza puste komórki we wskazanym zakresie.

Dobra rada

Bezargumentowa funkcja DZIS zwraca bieżącą datę. Jeżeli chcesz wstawić bieżącą datę do aktywnej komórki, możesz zastosować skrót **Ctrl + ;**.

Funkcja ROK • Rozwiązanie pierwszej części zadania wymaga użycia **funkcji ROK**. Zapiśmy w komórce H1 formułę `=ROK(B1)`. Aby ustalić numer kwartału, trzeba najpierw sprawdzić miesiąc, ponieważ miesiące należą do różnych kwartałów. Dlatego w komórce B3 chcemy uzyskać numer miesiąca, co zrealizujemy, stosując **funkcję MIESIĄC** w formule: `=MIESIĄC(B1)`.


Funkcja MIESIĄC • Formuła określająca kwartał będzie wymagała zagnieżdżenia funkcji: `=JEŻELI(B3<4;"I";JEŻELI(B3<7;"II";JEŻELI(B3<10;"III";"IV")))`

Określanie przyszłej daty

Pokażemy teraz, jak bez patrzenia w kalendarz określić datę wydarzenia w wybranym roku. Załóżmy, że w pewnej szkole rada pedagogiczna ustaliła dzień patrona szkoły na pierwszy czwartek marca każdego roku. Przyjmijmy, że w komórce F1 chcemy uzyskać datę dnia patrona szkoły w roku, który wpiszesz do komórki B1 (np. 2023).

Funkcja DZIEŃ.TYG • Najpierw niech program sprawdza, jakim dniem tygodnia jest 1 marca danego roku. W tym celu skorzystamy z **funkcji DZIEŃ.TYG**.

Dobra rada
Aby wprowadzić datę w formacie Excela, skorzystaj z funkcji **DATA**, której wynikiem jest odpowiednia liczba reprezentująca datę.



Rys. 22.9. Formuła z użyciem funkcji **DZIEŃ.TYG**

Pierwszy argument określa datę w formacie Excela, a drugi (opcjonalny) argument określa sposób numerowania dni tygodnia. Jeżeli pominiemy drugi argument lub wpiszesz 1, program zastosuje domyślne numerowanie: 1 – niedziela, 2 – poniedziałek, ..., 7 – sobota. Wpiszesz jako drugi parametr 14, co dla Excela oznacza numerowanie, według którego 1 to czwartek. Pierwszy argument (datę) określimy, stosując **funkcję DATA**.

Funkcja DATA •



Rys. 22.10. Formuła z użyciem funkcji **DATA**

Formułą określającą 1 marca danego roku będzie `=DATA(B1;3;1)`, ponieważ rok jest zapisany w komórce B1. Formuła w komórce B3, określająca dzień tygodnia, przyjmie zatem postać taką jak na rysunku 22.11: `=DZIEŃ.TYG(DATA(B1;3;1);14)`.

	A	B	C	D	E	F	G
1	Rok:	2023		Dzień patrona szkoły:		02.03.2023	
2							
3	Pierwszy marca:	7					
4							
5	Liczba dni do czwartku:	1					
6							

`=7-B3+1` `=DZIEŃ.TYG(DATA(B1;3;1);14)` `=DATA(B1;3;JEŻELI(B3=1;1;1+B5))`

Rys. 22.11. Wyznaczanie daty pierwszego czwartku marca

Jeżeli formuła w komórce B3 zwróci 1, będzie to oznaczało, że 1 marca w danym roku przypada w czwartek. Jeśli jednak formuła zwróci inną liczbę, musimy obliczyć liczbę dni do najbliższego czwartku. W tym celu w komórce B5 wpisujemy formułę $=7-B3+1$. $7-B3$ oznacza liczbę dni do pełnego tygodnia (w naszym przypadku do środy), dlatego dodajemy 1. Wówczas pierwszy czwartek marca, o ile nie jest pierwszym dniem, to dzień o numerze $1+B5$ (do pierwszego dnia marca dodajemy liczbę dni do najbliższego czwartku). Kompletna formuła w komórce F1 będzie zatem następująca: $=DATA(B1;3;JEŻELI(B3=1;1;1+B5))$.

Ćwiczenie 3

Zmodyfikuj powyższe rozwiązanie tak, aby nie korzystać z komórek pomocniczych. Zapisz odpowiednią formułę w komórce F3.

22.3. Funkcje tekstowe

W zależności od rodzaju danych wpisanych do komórek Excel automatycznie określa format tych danych. Komórki z formatem **Ogólne** mogą być traktowane m.in. jak liczbowe lub tekstowe. Jeżeli zawartość komórki rozpoczyna się znakiem = (równości), to Excel traktuje to jak początek formuły. Wyjątkiem jest sytuacja, gdy wpisujemy ciąg rozpoczynający się znakiem równości do komórki sformatowanej jako **Tekstowe**. Wówczas Excel interpretuje taki zapis dosłownie – wyświetla wpisany tekst.

Na podstawie rysunku 22.12 omówimy wybrane funkcje, które pomogą nam pracować z tekstem w arkuszu kalkulacyjnym. W komórce C1 wpisujemy imię i nazwisko (rozdzielone jedną spacją). Chcemy, aby w komórce F1 program zapisywał imię, w komórce F3 – nazwisko, a w komórce F5 – pierwszą literę imienia i nazwiska, czyli inicjały.

	A	B	C	D	E	F	G	H	I	J	K	L
1	Imię i nazwisko:		Jan Kowalski		Imię:	Jan		Pierwsza litera imienia:		J		
2												
3	Pozycja spacji:		4		Nazwisko:	Kowalski		Pierwsza litera Nazwiska:		K		
4												
5	Długość tekstu z C1:		12		Inicjały:	JK						
6												

$=ZNAJDŹ(" ";C1;1)$ $=LEWY(C1;C3-1)$ $=LEWY(F1)$ $=LEWY(F3)$
 $=DŁ(C1)$ $=ZŁĄCZ.TEKSTY(K1;K3)$ $=FRAGMENT.TEKSTU(C1;C3+1;C5-C3)$

Rys. 22.12. Wybrane funkcje tekstowe

Kluczowe jest to, żeby program znalazł pozycję spacji, która rozdziela imię i nazwisko. W tym celu w komórce C3 użyjemy funkcji **ZNAJDŹ**. • Funkcja ZNAJDŹ

Szukany tekst Przeszukiwany tekst Pozycja znaku, od którego rozpoczynamy wyszukiwanie

$=ZNAJDŹ(" ";C1;1)$

Rys. 22.13. Formuła z wykorzystaniem funkcji ZNAJDŹ

Dobra rada

Jeżeli chcesz, aby Excel traktował dane w komórce jako **Tekstowe**, rozpocznij wpisywanie tekstu od znaku ' (apostrofu).

Warto wiedzieć

W Excelu dostępna jest również funkcja SZUKAJ.TEKST, która w przeciwieństwie do ZNAJDŹ nie rozróżnia wielkości liter.

Funkcja DŁ • Pierwszy argument funkcji ZNAJDŹ określa szukany tekst. Drugi wskazuje tekst, który program ma przeszukiwać. Trzeci argument informuje, od której pozycji w tekście przeszukiwanym (od którego znaku) chcemy rozpocząć wyszukiwanie. Jest on opcjonalny – gdy go nie określimy, program przyjmie wartość domyślną, tj. 1.

Do wykonania zadania niezbędna będzie znajomość długości całego tekstu, co zrealizujemy za pomocą **funkcji DŁ**. Funkcja ta zlicza znaki tekstu będącego jej argumentem. W naszym przypadku formuła =DŁ(C1), którą wpisujemy w komórce C5, zwróci liczbę znaków w komórce C1. Aby wyświetlić imię, w komórce F1 zastosujemy

Dobra rada

Jeżeli pominiesz drugi argument w funkcji LEWY, to funkcja zwróci tylko jeden znak – pierwszy.

Tekst, z którego wyodrębniane będą znaki

Liczba znaków do wyodrębnienia z lewej strony

=LEWY(C1;C3-1)

Rys. 22.14. Formuła z użyciem funkcji LEWY

Formuła z rysunku 22.14 zwróci C3-1 początkowych znaków tekstu z komórki C1, zawierającej imię i nazwisko. W komórce C3 określona jest pozycja spacji rozdzielającej imię i nazwisko. Dlatego odejmujemy 1, by program nie dołączał spacji.

Funkcja FRAGMENT.TEKSTU •

Skorzystamy teraz z obliczonej już długości całego napisu, a także z **funkcji FRAGMENT.TEKSTU**, aby w komórce F3 wyświetlić nazwisko, czyli w naszym przypadku znaki znajdujące się po spacji (rys. 22.15).

Pozycja początkowa, od której wybieramy znaki

Komórka z tekstem zawierającym znaki, które mają zostać wyodrębnione

=FRAGMENT.TEKSTU(C1;C3+1;C5-C3)

Liczba znaków, które wybieramy

Rys. 22.15. Formuła z użyciem funkcji FRAGMENT.TEKSTU

Funkcja ZŁĄCZ.TEKSTY •

Warto wiedzieć

Funkcja ZŁĄCZ.TEKSTY może mieć maksymalnie 255 argumentów.

Aby uzyskać pierwszą literę imienia, stosujemy funkcję LEWY – tym razem możemy pominąć drugi argument. Analogicznie otrzymamy pierwszą literę nazwiska. Pozostaje nam połączyć te litery, stosując **funkcję ZŁĄCZ.TEKSTY** – łączy ona w jeden ciąg teksty określone w argumentach. W komórce F5 wpisujemy formułę: =ZŁĄCZ.TEKSTY(K1;K3)


Ćwiczenie 4

Stwórz arkusz, w którym w komórce A1 będziesz wpisywać imię i nazwisko, przy czym dopuszczamy wprowadzenie dwóch imion. Zastosuj takie formuły, aby: w komórce B1 wyświetlało się pierwsze imię, w komórce C1 – drugie imię, w komórce D1 – nazwisko, a w E1 – inicjał składający się z pierwszych liter obu imion i nazwiska. Rozwiązanie nie powinno zawierać komórek pomocniczych.

W wielu sytuacjach podczas analizowania danych tekstowych możemy chcieć szybko zliczyć, ile razy dany znak występuje w interesującym nas tekście. Choć Excel nie dysponuje funkcją przeznaczoną specjalnie do tego, możemy sprytnie wykorzystać inne dostępne w nim narzędzia.

Jak sprawdzić, ile znaków „a” zawiera tekst, który zapiszemy w wybranej komórce? Zadanie w gruncie rzeczy jest bardzo proste – wystarczy, że porównamy długość oryginalnego tekstu z długością tekstu bez litery „a”. Rysunek 22.16 pokazuje, jak to zrobić.

W komórce F1 za pomocą **funkcji PODSTAW** umieszczamy tekst z komórki B1 bez znaków „a”. Pierwszy argument funkcji PODSTAW określa tekst, który chcemy zmodyfikować. Drugi argument wskazuje tekst, który zamienimy na inny, zapisany w trzecim argumentcie.



Rys. 22.16. Formuła z użyciem funkcji PODSTAW

W komórkach B3 i F3 obliczamy długości tekstów: oryginalnego i bez litery „a”. W komórce J1 wyświetlamy różnicę długości obu tekstów, która odpowiada liczbie wystąpień znaku „a”.

	A	B	C	D	E	F	G	H	I	J	K
1	Napis:	baandaop		Napis bez "a":	bndop			Liczba liter "a":		3	
2											
3	Liczba znaków:	8		Liczba znaków bez "a":	5						
4											

`=DŁ.(B1)` `=DŁ.(F1)` `=PODSTAW(B1;"a";"")` `=B3-F3`

Rys. 22.17. Zliczanie wystąpień litery „a” w tekście znajdującym się w komórce B1

• Funkcja PODSTAW

Warto wiedzieć

Czwarty (opcjonalny) argument funkcji PODSTAW wskazuje, które wystąpienie szukanego znaku zamienimy innym. Jeżeli pominiemy ten argument, zamienione zostaną wszystkie wystąpienia szukanego znaku.

• Dobra rada

Aby zastąpić wybrany fragment jednego napisu innym, skorzystaj z funkcji ZASTAP.

A to ciekawe


Co potrafi arkusz kalkulacyjny?

Okazuje się, że arkusz kalkulacyjny to nie tylko tabele i formuły. Udowodnił to Japończyk Tatsuo Horiuchi, który w programie Microsoft Excel stworzył wiele pięknych rysunków. Zaczął malować w arkuszu kalkulacyjnym w 2000 r. Miał wówczas 60 lat. Japończyk wykorzystuje podstawowe narzędzia grafiki wektorowej, m.in. krzywe, i składa swoje obrazy z mniejszych elementów. Jego prace można obejrzeć np. na stronie <http://www2.odn.ne.jp/~cbl97790/>. Artysta udostępnił w internecie również instrukcje pokazujące, jak wykonać rysunki.



Ćwiczenie 5

Stwórz arkusz, w którym w komórce A1 będziesz zapisywać tekst składający się z ciągu małych liter bez spacji. Liczba znaków może wynosić od dwóch do ośmiu. W komórce B1 napisz formułę, która będzie wyświetlać ten sam ciąg znaków rozdzielonych spacjami.

Tabela programu Excel,
s. 334 


Dobra rada

Aby formuły były bardziej czytelne, ustal adekwatną do zawartości nazwę tabeli (**Projektowanie** → **Właściwości** → **Nazwa tabeli**).

22.4. Tabele i odwołania strukturalne

Przypomnijmy, że w Excelu możemy przekształcić zakres komórek w **tabelę**, która automatycznie dostosowuje swój rozmiar, kiedy dodamy albo usuniemy wiersze lub kolumny. Jeżeli chcemy w formule odwołać się do komórek z tabeli Excela, wystarczy podczas wpisywania formuły wskazać myszą odpowiedni zakres komórek. Rysunek 22.18 pokazuje tabelę o nazwie **Wymiary** i trzy przykłady odwołań do zawartości tabeli w formułach:

- ▶ odwołanie do całej tabeli, bez nazw kolumn (na rysunku oznaczone kolorem czerwonym) – wówczas podajemy tylko nazwę tabeli;
- ▶ odwołanie do kilku kolejnych kolumn tabeli (oznaczone kolorem pomarańczowym) – wówczas zapisujemy w nawiasie kwadratowym zakres (notacja z dwukropkiem), przy czym zamiast adresów komórek podajemy nazwy kolumn w nawiasach kwadratowych;
- ▶ odwołanie do innej komórki w tym samym wierszu arkusza (oznaczone kolorem zielonym) – wówczas nazwę kolumny poprzedzamy dodatkowo znakiem @ (małpa).



Rys. 22.18. Odwołania strukturalne

Odwołania strukturalne • Tego typu odwołania nazywamy **odwołaniami strukturalnymi**. Najłatwiej tworzyć je przez wskazywanie myszą odpowiednich zakresów komórek tabeli, ale można też wpisywać je ręcznie. Zamiast odwołań strukturalnych można stosować standardowy sposób adresowania. Na przykład na rysunku 22.18 równoważna formuła w komórce G5 może mieć postać =E5*F5.

Warto jednak używać odwołań strukturalnych, ponieważ zwiększają czytelność formuł. W dodatku formuła skopiowana do wszystkich komórek w wybranej kolumnie jest identyczna w całej kolumnie.

Ćwiczenie 6

Przygotuj tabelę według rysunku 22.18 i sprawdź, czy formuła z komórki G5 będzie identyczna po skopiowaniu do pozostałych komórek w tej kolumnie tabeli.

Łączenie danych z kilku tabel

Wyobraź sobie, że z okazji 10. rocznicy wydania gry fantasy przygotowano dla graczy niespodziankę. Każdy gracz otrzyma dodatkowe punkty doświadczenia (na podstawie już posiadanych) oraz unikatowy przedmiot (inny dla każdej klasy postaci). Rysunek 22.19 przedstawia trzy tabele. Tabela Gracze zawiera listę uczestników gry, liczbę zdobytych przez nich punktów oraz klasę postaci, do której należą. Tabele Przedmioty i Doświadczenie określają zasady przyznawania jubileuszowych nagród.

Tabela Gracze						Tabela Przedmioty		Tabela Doświadczenie	
ID gracza	Pseudonim gracza	Punkty	Klasa postaci	Przedmiot	Bonus	Klasa postaci	Przedmiot	Przedmiot	Bonus
4046	spytynKaptanka	326	Kaptanka			Czarodziejka	pieńczak	100	300
3446	odważnyMag	828	Mag			Mag	amulet	400	200
1611	odważnaKaptanka	108	Kaptanka			Krasnolud	topór	700	100
3757	niepokonanyKrasnolud	630	Krasnolud			Kaptanka	szyflet	900	50
3698	spytynMag	501	Mag			Wojownik	miecz		
2590	odważnaCzarodziejka	320	Czarodziejka						
2594	odważnyWojownik	264	Wojownik						
3710	odważnaCzarodziejka	513	Czarodziejka						
2812	niepokonanyKrasnolud	186	Krasnolud						
1548	spytynMag	375	Mag						
2957	niepokonanyKrasnolud	251	Krasnolud						
1177	spytynWojownik	704	Wojownik						
1159	odważnyMag	610	Mag						
2444	niepokonanaKaptanka	746	Kaptanka						
2965	odważnaKaptanka	996	Kaptanka						
1120	niepokonanaKaptanka	600	Kaptanka						
1621	odważnyWojownik	579	Wojownik						
1552	spytynKrasnolud	244	Krasnolud						
3681	niepokonanyWojownik	190	Wojownik						
1475	odważnyKrasnolud	435	Krasnolud						

Rys. 22.19. Tabele w programie Excel

Naszym zadaniem jest uzupełnienie kolumn Przedmiot i Bonus w tabeli Gracze danymi z tabel Przedmioty i Doświadczenie. W tym celu skorzystamy z funkcji WYSZUKAJ.PIONOWO. Formuła z jej użyciem może być taka jak na rysunku 22.20.

Szukana wartość

=WYSZUKAJ.PIONOWO(D2;H2:I6;2;FAŁSZ)

Obszar poszukiwania wartości (zawsze pierwsza kolumna z zakresu)

Numer kolumny w zakresie, z którego będzie zwracana wartość

Sposób dopasowania wartości (FAŁSZ oznacza dokładny, PRAWDA – przybliżony)

Rys. 22.20. Formuła z użyciem funkcji WYSZUKAJ.PIONOWO

Dobra rada

Aby obszar poszukiwań nie zmieniał się po skopiowaniu formuły, zastosuj adresowanie bezwzględne, np. \$H\$2:\$I\$6.

Warto wiedzieć

Funkcja WYSZUKAJ.PIONOWO znajduje wyłącznie pierwsze wystąpienie szukanej wartości. Poza tym nie rozróżnia wielkości znaków, tzn. napisy „PDF” i „pdf” traktuje jak identyczne.

Dobra rada

Ostatni parametr funkcji WYSZUKAJ.PIONOWO jest opcjonalny, domyślnie przyjmuje wartość PRAWDA. Możesz stosować skrócony zapis: 0 zamiast FAŁSZ i 1 zamiast PRAWDA.

W komórce E2 wpisujemy początek formuły: =WYSZUKAJ.PIONOWO(i zaznaczamy myszą komórkę D2 – w **Pasku formuły** pojawi się napis @[Klasa postaci]]. Następnie wprowadzamy ; (średnik) i wskazujemy myszą zakres H2:I6. W **Pasku formuły** pojawi się napis Przedmioty (nazwa tabeli). Dopisujemy pozostałą część formuły: ;2;FAŁSZ) i zatwierdzamy, naciskając **Enter**. Kompletna formuła przyjmuje postać: =WYSZUKAJ.PIONOWO([@Klasa postaci]);Przedmioty;2;FAŁSZ)

Oznacza to, że program będzie szukać wartości z bieżącej wiersza kolumny Klasa postaci. Przeszukiwany zakres to pierwsza kolumna tabeli Przedmioty. Ostatni argument (FAŁSZ) określa dokładne dopasowanie, tzn. porównywane są po kolei wszystkie wiersze, aż do znalezienia szukanej wartości lub do wyczerpania przeszukiwanego zakresu.

Jeżeli szukana wartość zostanie znaleziona, to funkcja zwraca wartość z tego wiersza oraz kolumny określonej w trzecim parametrze, w naszym przypadku z drugiej kolumny (Przedmiot). W przeciwnym razie w wyniku otrzymamy kod błędu #N/D!. Aby określić, które przedmioty otrzymają pozostali gracze, wystarczy skopiować formułę z komórki E2 do końca kolumny tabeli Gracze. Po skopiowaniu formuła jest identyczna we wszystkich komórkach zakresu E2:E21, ponieważ odwołanie @[Klasa postaci] zawsze oznacza wiersz tabeli, w którym to odwołanie zostało użyte.

Formuła, dzięki której wyznaczymy liczbę dodatkowych punktów doświadczenia, jest analogiczna: =WYSZUKAJ.PIONOWO([@Punkty]);Doświadczenie;2;PRAWDA)

Wpiszemy ją w kolumnie Bonus do komórki F2. Istotną zmianę w działaniu funkcji WYSZUKAJ.PIONOWO wprowadza ostatni argument (PRAWDA) – określa dopasowanie przybliżone. Wyszukiwanie z dopasowaniem przybliżonym jest dużo szybsze niż wyszukiwanie z dopasowaniem dokładnym. Aby zastosować to dopasowanie, dane w przeszukiwanej kolumnie (w pierwszej kolumnie tabeli Doświadczenie) muszą być posortowane rosnąco. Wynika to z faktu, że wyszukiwanie odbywa się według algorytmu wyszukiwania binarnego. Skopiujemy formułę z komórki F2 aż do komórki F21, aby ukończyć zadanie.

Ćwiczenie 7

W zakresie A1:B11 przygotuj zestawienie 10 państw europejskich wraz ze stolicami. Następnie utwórz formułę, która w komórce E1 wyświetli nazwę stolicy państwa z komórki D1.

Przeliczanie walut

Wyobraźmy sobie, że wyruszamy w podróż przez wiele krajów. Postanawiamy, że w każdym odwiedzonym kraju będziemy płacić wyłącznie w walucie miejscowej. Za każdym razem zaraz po przybyciu z kraju A

do kraju B będziemy zamieniać walutę z A na B. Zakładamy, że prowizja przy wymianie waluty wyniesie 15% wymienianej kwoty. Najpierw chcemy jednak sprawdzić na podstawie kursów podawanych przez Narodowy Bank Polski, ile zapłacimy za daną kwotę w walucie B (z uwzględnieniem prowizji za wymianę). Do wykonania tych obliczeń wygodnie będzie użyć arkusza kalkulacyjnego.

Najpierw **importujemy** aktualną tabelę A kursów średnich z serwisu Narodowego Banku Polskiego. Zimportowane dane są już sformatowane jako tabela programu Excel. Zmieniamy nazwę tabeli na **KursyNBP**.

W kolejnym kroku przygotowujemy komórki do wprowadzania danych. Możemy to zrobić np. tak jak na rysunku 22.21.


Importowanie danych ze źródeł internetowych,
s. 324–327 

	A	B	C	D	E	F	G	H	I	J	K
1	Kwota w walucie A:	100		Szacowana kwota w walucie B:			1060,284			0,1675	
2	Kod waluty A:	1 CZK								0,013428	
3	Kod waluty B:	100 HUF									
4											
5	Nazwa waluty	Kod waluty	Kurs średni								
6	bat (Tajlandia)	1 THB	0,1152								
7	dolar amerykański	1 USD	3,7804								
8	dolar australijski	1 AUD	2,7383								
9	dolar Hongkongu	1 HKD	0,483								
10	dolar kanadyjski	1 CAD	2,8949								
11	dolar nowozelandzki	1 NZD	2,5219								
12	dolar singapurski	1 SGD	2,7557								
13	euro	1 EUR	4,3244								
14	forint (Węgry)	100 HUF	1,3428								
15	frank szwajcarski	1 CHF	3,7825								

Rys. 22.21. Propozycja wprowadzenia danych

Przyjmujemy, że kwotę w walucie kraju A wprowadzamy do komórki B1. W komórce G1 mamy otrzymać szacowaną kwotę w walucie B. Aby uniknąć literówek, kody walut A i B będziemy wybierać z listy rozwijanej. Wyjaśnijmy sposób tworzenia listy wyboru na przykładzie kodu waluty A. Najpierw zaznaczamy komórki zawierające wartości, które mają być dostępne na liście rozwijanej (B6:B40). Następnie wybieramy **Formuły** → **Nazwy zdefiniowane** → **Definiuj nazwę** i nadajemy zaznaczonym komórkom nazwę kw. Utworzyliśmy w ten sposób zakres, do którego będziemy mogli się odwoływać, podając jego nazwę zamiast adresu komórek.

Następnie zaznaczamy komórkę B2 i wybieramy **Dane** → **Narzędzia danych** → **Poprawność danych** i ustawiamy parametry w sposób przedstawiony na rysunku 22.22. W ten sam sposób określamy sprawdzanie poprawności dla kodu waluty B (komórka B3).



Rys. 22.22. Ustawienia listy wyboru

Dobra rada

Aby sprawdzić listę nazwanych zakresów, rozwiń **Pole nazwy**, znajdujące się po lewej stronie **Paska formuły**. Nazwami możesz zarządzać po wybraniu **Formuły** → **Nazwy zdefiniowane** → **Menedżer nazw**.

Zgodnie z warunkami określonymi w zadaniu kurs będziemy przeliczać według wzoru:

$$0,85 \cdot \text{kwota w walucie A} \cdot \frac{\text{kurs waluty A}}{\text{kurs waluty B}}$$

Współczynnik 0,85 uwzględnia 15-procentową prowizję pobieraną przez bank. Dodatkowa trudność polega na tym, że w przypadku niektórych walut NBP podaje kurs za 100 lub więcej jednostek tej waluty. Kod waluty poprzedzony jest odpowiednią liczbą, a po spacji zapisany jest trzyliterowy symbol waluty. Warto zatem w komórce pomocniczej, np. J1, obliczyć kurs waluty A w przeliczeniu na jedną jednostkę waluty. Określi go formuła utworzona według wzoru:

$$\text{kurs waluty A} = \frac{\text{kurs średni NBP}}{\text{wsp. A}}$$

Współczynnik (wsp. A) oznacza tu liczbę poprzedzającą symbol waluty, np. w przypadku forinta węgierskiego wsp. A wynosi 100. Postać formuły w komórce J1 jest następująca:

```
=WYSZUKAJ.PIONOWO(B2;KursyNBP[[Kod waluty ]:  
[Kurs średni]];2;0)/(1*LEWY(B2;ZNAJDŹ(" ";B2)))
```

Funkcja LEWY,
s. 362 

Zwróć uwagę, że **funkcja LEWY** pozwala znaleźć tekstową postać współczynnika, który najłatwiej zamienić na liczbę, mnożąc go przez jeden. W podobny sposób w komórce J2 obliczymy kurs waluty B:

```
=WYSZUKAJ.PIONOWO(B3;KursyNBP[[Kod waluty ]:  
[Kurs średni]];2;0)/(1*LEWY(B3;ZNAJDŹ(" ";B3)))
```

Na zakończenie w komórce G1 obliczymy ostateczny wynik z uwzględnieniem wszystkich obliczeń pośrednich i warunków zadania. Formuła w tej komórce przyjmie postać: $=0,85*B1*J1/J2$.

Zapamiętaj

Funkcja WYSZUKAJ.PIONOWO umożliwia wyszukiwanie i łączenie danych z różnych wierszy według wybranego klucza. Gdy stosujemy funkcję WYSZUKAJ.PIONOWO z dopasowaniem przybliżonym, przeszukiwany zakres musi zawierać dane posortowane rosnąco.

Ćwiczenie 8


Na podstawie tabeli A kursów średnich z serwisu Narodowego Banku Polskiego wykonaj następujące zadania.

- Zastosuj opisane powyżej rozwiązania do wykonania arkusza według rysunku 22.21 ze s. 367.
- Zmodyfikuj formułę w komórce G1, aby nie odwoływała się do komórek pomocniczych J1 oraz J2.

22.5. Formuły tablicowe

Formuły, których używaliśmy do tej pory, zwracają wynik do pojedynczej komórki. Jeżeli chcemy w wyniku otrzymać więcej wartości, korzystamy z **formuł tablicowych**, które umożliwiają wykonywanie operacji na wielu elementach tablicy. Każdą formułę możemy potraktować jako formułę tablicową. Przekonamy się o tym, rozwiązując następujący problem.

Chcemy określić, który ciąg znaków – „UGA” czy „UAA” – występuje wcześniej w ciągu „UUGCAAUGACGCUAA”. Zastosujemy funkcję ZNAJDŹ. Jeżeli wskażemy jako szukany tekst zakres komórek zawierający oba szukane ciągi, formuła =ZNAJDŹ(B3:B4;B1) zwróci błąd #ARG! – niewłaściwy argument funkcji (komórka I1 na rysunku 22.23).



Rys. 22.23. Zastosowanie formuł tablicowych

Jeżeli jednak zaznaczymy zakres I3:I4 i tę samą formułę wprowadzimy jako formułę tablicową (zatwierdzimy, naciskając kombinację **Ctrl + Shift + Enter**), w zaznaczonych komórkach otrzymamy dwie liczby – tak jakbyśmy użyli dwóch formuł. Ponieważ jako pierwszy parametr funkcji ZNAJDŹ podaliśmy zakres dwuelementowy (B3:B4), formuła tablicowa zwraca również tablicę dwuelementową. Możemy się o tym przekonać, kiedy spróbujemy wprowadzić formułę tablicową do jednej komórki albo do trzech komórek. Gdy zakres wyjściowy jest mniejszy, zwracany jest tylko fragment tablicy (komórka K3 na rysunku powyżej), a w komórkach przekraczających zakres wejściowy zwracany jest błąd #N/D! (komórka L5).

Kiedy program zna już pozycje obydwu ciągów w tekście, wystarczy, że wybierze mniejszą liczbę. Skorzystamy w tym celu z **funkcji MIN**. Zwraca ona najmniejszą liczbę, argumentami mogą być liczby, tablice, nazwy i odwołania zawierające liczby. Kompletna formuła w komórce E7 ma zatem postać =MIN(ZNAJDŹ(B3:B4;B1)). Formuła tablicowa w **Pasku formuły** wyświetlana jest w nawiasie klamrowym.

Ćwiczenie 9

Za pomocą formuły tablicowej oblicz kwadraty liczb zapisanych w komórkach A1:A5.

Formuła tablicowa

Warto wiedzieć

W biologii molekularnej UGA, UAG oraz UAA to tzw. kodony terminacyjne, które kodują zakończenie biosyntezy.

Warto wiedzieć

Formuły tablicowe nazywane są również formułami CSE, od pierwszych liter klawiszy, za pomocą których się je wprowadza:

Ctrl + Shift + Enter.

Funkcja MIN

Stałe tablicowe

Oceny to tabela utworzona w arkuszu zawierająca zapis liczbowy i słowny ocen szkolnych (rys. 22.24). Formuła `=WYSZUKAJ.PIONOWO(A1;Oceny;2;0)` zwróci postać słowną oceny zapisanej liczbą w komórce A1.

Można jednak zrezygnować z umieszczenia tabeli z ocenami w komórkach arkusza i zapisać ją bezpośrednio w formule:

```
=WYSZUKAJ.PIONOWO(A1;{1\"niedostateczny\";
2\"dopuszczający\";3\"dostateczny\";4\"dobry\";
5\"bardzo dobry\";6\"celujący\"};2;0)
```

Ocena	Słownie
1	niedostateczny
2	dopuszczający
3	dostateczny
4	dobry
5	bardzo dobry
6	celujący

Rys. 22.24. Tabela z ocenami

Stała tablicowa • Tabelę taką zapisujemy w nawiasie klamrowym {} i nazywamy **stałą tablicową**. Wiersze tabeli rozdzielone są znakiem ; (średnik), wartości w wierszu rozdziela znak \ (lewy ukośnik, ang. *backslash*). Zwróć uwagę, że nie trzeba powyższej formuły wpisywać jako formuły tablicowej.

Jeśli do komórki A1 wpisujemy 2, formuła zwróci tekst „dopuszczający”. Formuła zwróci kod błędu #N/D!, jeśli liczby z komórki A1 nie ma w pierwszej kolumnie przeszukiwanej tabeli. Najprościej jest zabezpie-

Funkcja JEŻELI.BŁĄD • czyć się przed takim błędem, stosując **funkcję JEŻELI.BŁĄD** w sposób pokazany na rysunku 22.25.

```
=JEŻELI.BŁĄD(WYSZUKAJ.PIONOWO(A1;{1\"niedostateczny\";2\"
dopuszczający\";3\"dostateczny\";4\"dobry\";5\"
bardzo dobry\";6\"celujący\"};2;0);\"")
```

Obliczane wyrażenie

Wynik zwracany przez formułę, gdy wyrażenie zwróci błąd

Rys. 22.25. Formuła z użyciem funkcji JEŻELI.BŁĄD

Pierwszy argument (zaznaczony na zielono) to dowolne wyrażenie, które może zwrócić kod błędu. Drugi argument (zaznaczony na czerwono) to wynik formuły, gdy wyrażenie zwróci kod błędu.

Dobra rada

Jeżeli formuła w arkuszu nie może zostać obliczona, program zwraca kod błędu, np.:

#ARG!, #N/D!

Informacje o kodach błędów możesz znaleźć w pomocy programu.

Zapamiętaj

Formuły tablicowe rozszerzają możliwości funkcji tak, że mogą one działać na zakresach (przyjmować je jako parametry oraz zwracać w wyniku działania). Formuły tablicowe zatwierdzamy, naciskając kombinację **Ctrl + Shift + Enter**.

Ćwiczenie 10

Porównaj opisane powyżej sposoby zamiany oceny liczbowej na zapis słowny ze sposobem, w którym wykorzystasz zagnieżdżoną funkcję JEŻELI. Określ zalety i wady tych rozwiązań.

Podsumowanie

- Zagnieżdżanie funkcji pozwala tworzyć złożone formuły.
- W programie Excel dostępne są funkcje umożliwiające wykonywanie podstawowych operacji na tekstach (np. ZNAJDŹ, FRAGMENT.TEKSTU) oraz na datach (np. DZIEŃ.TYG, DATA).
- Odwołania strukturalne w tabelach zwiększają czytelność formuł.
- Funkcja WYSZUKAJ.PIONOWO umożliwia łączenie danych z różnych wierszy na podstawie wspólnej wartości. Domyślnie stosowane jest wyszukiwanie przybliżone.
- Każdą formułę można wprowadzić jako tablicową za pomocą klawiszy **Ctrl + Shift + Enter**. Dzięki temu funkcje użyte w formule tablicowej mogą przyjmować jako argumenty całe zakresy, mogą również w wyniku zwracać tablice.
- Formuły tablicowe wyświetlane są w **Pasku formuły** w nawiasach klamrowych.

Zadania

- ★ **1** Dla daty wprowadzonej do komórki A1 zaproponuj w komórce B1 formułę, która wypisze słownie dzień tygodnia.
- ★ **2** W komórce B1 arkusza wstaw formułę, która zwraca „Tak”, gdy liczba zapisana w komórce A1 jest podzielna przez 2 lub 3.
- ★ **3** W zakresie komórek C2:C10 zapisz formułę tablicową, która zwróci iloczyn liczb z dwóch sąsiednich komórek z kolumn A i B.
- ★ **4** W komórce A1 zapisz dowolne słowo. W komórce B1 zapisz formułę, która zwróci pierwszą i ostatnią literę słowa z komórki A1, np. dla słowa „kot” zwróci „kt”.
- ★★ **5** Oblicz, ile razy tekst „ab” występuje w ciągu „babccbabcacababcabbbabc”. W rozwiązaniu nie stosuj odwołań do innych komórek z obliczeniami pośrednimi.
- ★★ **6** Załóżmy, że rozszerzasz szkolny system oceniania o dodatkowe oceny: 7 – „znakomity”, 8 – „doskonały”, 9 – „wyśmienity”, 10 – „wybitny”. W komórce B1 zapisz formułę, która zwróci zapis słowny oceny z komórki A1 (zapisanej liczbą) w rozszerzonym systemie oceniania.
- ★★ **7** Zaprojektuj tabelę Excela zawierającą kolumny: „Rodzaj płatności”, „Data płatności” oraz „Czas na opłatę”. W pierwszej kolumnie wpisz, czego dotyczy płatność, np. czynsz, internet. W drugiej kolumnie zapisz termin płatności, a w ostatniej – formułę, która wyświetli liczbę dni, po których mija termin płatności.
- ★★ **8** W komórce A1 wpisz formułę, która zwróci aktualną liczbę dni nauki pozostałych do końca roku szkolnego. Przyjmij, że pierwszym dniem wakacji jest 1 lipca i nie ma świąt przypadających w dni robocze. Skorzystaj z funkcji DNI.ROBOCZE.

*** 9 W komórce A1 zapisz dowolną liczbę całkowitą dodatnią. Zaproponuj formułę, która zwróci „Tak”, jeśli liczba z komórki A1 jest palindromiczna, oraz „Nie” – w przeciwnym wypadku.

*** 10 Za pomocą formuł tablicowych wykonaj tabliczkę mnożenia w układzie ósemkowym.

	1	2	3	4	5	6	7
1	1	2	3	4	5	6	7
2	2	4	6	10	12	14	16
3	3	6	11	14	17	22	25
4	4	10	14	20	24	30	34
5	5	12	17	24	31	36	43
6	6	14	22	30	36	44	52
7	7	16	25	34	43	52	61

*** 11 Przygotuj arkusz, w którym w komórce C1 wyświetli się „Tak”, jeżeli słowo zapisane w komórce A1 jest nazwą powiatu, i „Nie”, jeżeli nie istnieje powiat o takiej nazwie. Listę powiatów pobierz z bazy TERC z serwisu Głównego Urzędu Statystycznego.

*** 12 Wyjaśnij sposób działania formuły widocznej na rysunku. Formuła ta zwraca sumę cyfr liczby naturalnej wpisanej do komórki B1.

F1		: X ✓ f_x		={SUMA(1*FRAGMENT.TEKSTU(B1;WIERSZ(ADR.POŚR("1."&DŁ{B1}));1))}							
	A	B	C	D	E	F	G	H	I	J	K
1	Liczba:	23455667			Suma cyfr:	38					
2											